McCabe-Thiele analysis for a pinch at top of column as a function of α.

This notebook has been written in *Mathematica* by

Mark J. McCready
Professor of Chemical and Biomolecular Engineering
University of Notre Dame
Notre Dame, IN 46556
USA

mjm@nd.edu
http://chemeprof.com

Version: 6/29/2021

This notebook addresses how the expected number of trays (stages) would change for a tough separation as the relative volatility is varied in a range of values slightly above 1.

Using α as the relative volatility, the number of trays varies as $1/(\alpha-1)$, or 1/Log[α], which are about the same for α close to 1. This is assuming constant α and is for the limit infinite reflux ratio. This result, with the same α scaling, is similar to the Fenske equation (King, Separation Processes, 1971, p 456) which is derived for constant α and total reflux but not near a pinch.

Equilibrium relation

The definition of relative volatility is

```
eq1 = y1 / x1 / ( y2 / x2) == α
```

$$\frac{x2\ y1}{x1\ y2} == \alpha$$

```
eq2 = eq1 /. {y2 → 1 - y1, x2 → 1 - x1}
```

$$\frac{(1 - x1)\, y1}{x1\, (1 - y1)} == \alpha$$

For this analysis we want to linearize this expression around x1=1.

```
why1 = y1 /. Solve[eq2, y1][[1]]
```

$$\frac{x1\, \alpha}{1 - x1 + x1\, \alpha}$$

```
whyexpand = Series[why1, {x1, 1, 1}]
```

$$1 + \frac{x1 - 1}{\alpha} + O[x1 - 1]^2$$

Let's check the expansion:

$$\text{Plot}\left[\left\{\left(\frac{x1\, \alpha}{1 - x1 + x1\, \alpha}\right) /.\ \alpha \to 2,\ \left(1 + \frac{x1 - 1}{\alpha}\right) /.\ \alpha \to 2\right\}, \{x1, 0, 1\}\right]$$



The range of linear behavior is (obviously) wider if α is closer to 1.

```
Plot[{(x1 α/(1 - x1 + x1 α)) /. α → 1.1, (1 + (x1 - 1)/α) /. α → 1.1}, {x1, 0, 1}]
```




```
eqgraph =
 Plot[{(x1 α/(1 - x1 + x1 α)) /. α → 1.3, (1 + (x1 - 1)/α) /. α → 1.3}, {x1, .6, 1}]
```



So the equilibrium line at the top is y = $1/\alpha$ + $1-1/\alpha$

operating line

```
op1 = y - y0 == L / G ( x - x0)
```

$$y - y0 == \frac{L\ (x - x0)}{G}$$

```
whyop = y /. Solve[op1, y][[1]]
```

$$\frac{L\ x - L\ x0 + G\ y0}{G}$$

Since y0=x0

```
whyop1 = whyop /. x0 → y0
```

$$\frac{L\ x + G\ y0 - L\ y0}{G}$$

```
opgraph =
 Plot[whyop1 /. {y0 → .99, L → .95, G → 1}, {x, .6, .99}, PlotStyle → Red]
```



```
Show[eqgraph, opgraph]
```



So there is some accessible range for the linear approximation. It looks like the L/G will need to be pretty close to unity if α is this small.

Infinite reflux ratio

For an infinite reflux ratio, each step "starts"on the 45 degree line, (y=x). This makes calculating successive steps just a matter of using a simple recursion relation. Each step is just a "move" at constant y from the y=x line to the left until the x value of the intersection with the equilibrium line is found.

To do this we solve the equilibrium relation for x

$$\text{exeq} = x \;/.\; \text{Solve}\left[y == 1 + \frac{x-1}{\alpha}, x\right][[1]]$$

$1 - \alpha + y\,\alpha$

To do the stepping, we would start at y = y0 (say =.99) and calculate x. For α = 1.3 we get

```
exeq /. {y → .99, α → 1.3}
```

0.987

Mathematica can apply a function recursively using the “NestList” command. Let’s do 5 steps.

```
exp1 = NestList[1 - α + α # &, y0, 5]
```

```
{y0, 1 - α + y0 α, 1 - α + α (1 - α + y0 α),
 1 - α + α (1 - α + α (1 - α + y0 α)), 1 - α + α (1 - α + α (1 - α + α (1 - α + y0 α))),
 1 - α + α (1 - α + α (1 - α + α (1 - α + α (1 - α + y0 α))))}
```

```
exp2 = exp1 /. {y0 → .99, α → 1.3}
```

```
{0.99, 0.987, 0.9831, 0.97803, 0.971439, 0.962871}
```

To make a plot lets try 10 steps. We want to mimic the McCabe - Thiele graphical construction.

```
numbs = (NestList[1 - α + α # &, y0, 10]) /. {y0 → .99, α → 1.3}
```

```
{0.99, 0.987, 0.9831, 0.97803, 0.971439, 0.962871,
 0.951732, 0.937251, 0.918427, 0.893955, 0.862142}
```

Make a list of pairs of these numbers for the 45 line intersections:

```
ans1 = Transpose[{numbs, numbs}]
```

```
{{0.99, 0.99}, {0.987, 0.987}, {0.9831, 0.9831},
 {0.97803, 0.97803}, {0.971439, 0.971439}, {0.962871, 0.962871},
 {0.951732, 0.951732}, {0.937251, 0.937251},
 {0.918427, 0.918427}, {0.893955, 0.893955}, {0.862142, 0.862142}}
```

Now get the eq line intersections which are pairs of a the previous “45” value as y and the new “x” from the 1-$\alpha + \alpha$ y function. You might have to trust me!

```
ans2 = Transpose[
  { Take[numbs, {2, Length[numbs]}], Take[numbs, Length[numbs] - 1]
        }]
```

```
{{0.987, 0.99}, {0.9831, 0.987}, {0.97803, 0.9831},
 {0.971439, 0.97803}, {0.962871, 0.971439}, {0.951732, 0.962871},
 {0.937251, 0.951732}, {0.918427, 0.937251},
 {0.893955, 0.918427}, {0.862142, 0.893955}}
```

Now get the eq and “45” pairs alternating in 1 list for plotting:

```
ans3 = Flatten[Table[{ans1[[i]], ans2[[i]]}, {i, 1, Length[ans2]}], 1]
```

```
{{0.99, 0.99}, {0.987, 0.99}, {0.987, 0.987},
 {0.9831, 0.987}, {0.9831, 0.9831}, {0.97803, 0.9831},
 {0.97803, 0.97803}, {0.971439, 0.97803}, {0.971439, 0.971439},
 {0.962871, 0.971439}, {0.962871, 0.962871},
 {0.951732, 0.962871}, {0.951732, 0.951732}, {0.937251, 0.951732},
 {0.937251, 0.937251}, {0.918427, 0.937251}, {0.918427, 0.918427},
 {0.893955, 0.918427}, {0.893955, 0.893955}, {0.862142, 0.893955}}
```

One more point at the bottom

```
ans4 = ans3[[Length[ans3]]][[2]]
```

```
0.893955
```

Put this one into the list

```
steps = Join[ans3, {{ans4, ans4}}]
```

```
{{0.99, 0.99}, {0.987, 0.99}, {0.987, 0.987},
 {0.9831, 0.987}, {0.9831, 0.9831}, {0.97803, 0.9831},
 {0.97803, 0.97803}, {0.971439, 0.97803}, {0.971439, 0.971439},
 {0.962871, 0.971439}, {0.962871, 0.962871}, {0.951732, 0.962871},
 {0.951732, 0.951732}, {0.937251, 0.951732}, {0.937251, 0.937251},
 {0.918427, 0.937251}, {0.918427, 0.918427}, {0.893955, 0.918427},
 {0.893955, 0.893955}, {0.862142, 0.893955}, {0.893955, 0.893955}}
```

Plot this list which is the steps:

```
stepplot = ListPlot[steps, Joined → True]
```



Plot the equilibrium line

```
eqline = Plot[(1 + (x - 1)/α) /. α → 1.3, {x, .86, 1}, PlotStyle → Green]
```



plot y = x

```
whyex = Plot[x, {x, .86, 1}, PlotStyle → Red]
```




```
fullresult = Show[stepplot, whyex, eqline]
```



We see that the 10 steps get a mixture at ~.86 up to .99.

■ How will this change with α?

To make a plot lets try 10 steps. We want to mimic the McCabe - Thiele graphical construction.

```
numbs = (NestList[1 - α + α # &, y0, 10]) /. {y0 → .99, α → 1.2}
```

```
{0.99, 0.988, 0.9856, 0.98272, 0.979264, 0.975117,
 0.97014, 0.964168, 0.957002, 0.948402, 0.938083}
```

Make a list of pairs of these numbers for the 45 line intersections:

```
ans1 = Transpose[{numbs, numbs}]
```

```
{{0.99, 0.99}, {0.988, 0.988}, {0.9856, 0.9856},
 {0.98272, 0.98272}, {0.979264, 0.979264}, {0.975117, 0.975117},
 {0.97014, 0.97014}, {0.964168, 0.964168}, {0.957002, 0.957002},
 {0.948402, 0.948402}, {0.938083, 0.938083}}
```

Now get the eq line intersections which are pairs of a the previous “45” value as y and the new “x” from the 1-$\alpha + \alpha$ y function. You might have to trust me!

```
ans2 = Transpose[
  { Take[numbs, {2, Length[numbs]}], Take[numbs, Length[numbs] - 1]
          }]
```

```
{{0.988, 0.99}, {0.9856, 0.988}, {0.98272, 0.9856},
 {0.979264, 0.98272}, {0.975117, 0.979264},
 {0.97014, 0.975117}, {0.964168, 0.97014}, {0.957002, 0.964168},
 {0.948402, 0.957002}, {0.938083, 0.948402}}
```

Now get the eq and “45” pairs alternating in 1 list for plotting:

```
ans3 = Flatten[Table[{ans1⟦i⟧, ans2⟦i⟧}, {i, 1, Length[ans2]}], 1]
```

```
{{0.99, 0.99}, {0.988, 0.99}, {0.988, 0.988},
 {0.9856, 0.988}, {0.9856, 0.9856}, {0.98272, 0.9856},
 {0.98272, 0.98272}, {0.979264, 0.98272}, {0.979264, 0.979264},
 {0.975117, 0.979264}, {0.975117, 0.975117}, {0.97014, 0.975117},
 {0.97014, 0.97014}, {0.964168, 0.97014}, {0.964168, 0.964168},
 {0.957002, 0.964168}, {0.957002, 0.957002},
 {0.948402, 0.957002}, {0.948402, 0.948402}, {0.938083, 0.948402}}
```

One more point at the bottom

```
ans4 = ans3⟦Length[ans3]⟧⟦2⟧
```

```
0.948402
```

Put this one into the list

```
steps = Join[ans3, {{ans4, ans4}}]
```

```
{{0.99, 0.99}, {0.988, 0.99}, {0.988, 0.988},
 {0.9856, 0.988}, {0.9856, 0.9856}, {0.98272, 0.9856},
 {0.98272, 0.98272}, {0.979264, 0.98272}, {0.979264, 0.979264},
 {0.975117, 0.979264}, {0.975117, 0.975117}, {0.97014, 0.975117},
 {0.97014, 0.97014}, {0.964168, 0.97014}, {0.964168, 0.964168},
 {0.957002, 0.964168}, {0.957002, 0.957002}, {0.948402, 0.957002},
 {0.948402, 0.948402}, {0.938083, 0.948402}, {0.948402, 0.948402}}
```

Plot this list which is the steps:

```
stepplot = ListPlot[steps, Joined → True]
```



Plot the equilibrium line

eqline = Plot$\left[\left(1+\frac{x-1}{\alpha}\right)$ /. $\alpha \to 1.2$, {x, .86, 1}, PlotStyle → Green$\right]$



plot y = x

```
whyex = Plot[x, {x, .86, 1}, PlotStyle → Red]
```




```
fullresult = Show[stepplot, whyex, eqline]
```



We see that the 10 steps get a mixture at ~.93up to .99 if α is reduced.

■ Can we pick a range and α and solve for the needed stages?

```
ExpandAll[NestList[1 - α + α # &, y0, 5]]
```

$\{y0,\ 1-\alpha+y0\,\alpha,\ 1-\alpha^2+y0\,\alpha^2,\ 1-\alpha^3+y0\,\alpha^3,\ 1-\alpha^4+y0\,\alpha^4,\ 1-\alpha^5+y0\,\alpha^5\}$

We see that there is a starting composition that would work for any chosen number of stages. For, say 5 stages, if we started with x = .975, we could get to .99.

$1-\alpha^5+y0\,\alpha^5$ /. {y0 → .99, α → 1.2}

```
0.975117
```

Since the result of the recurring function is simple, we can pick a desired starting mixture and calculate the number stages to get to, say, y=0.99.

We just the value of “n” after we pick α, x0 and a starting x. Let’s start at .86 to match the first example.

```
FindRoot[((1 - α^n + x0 α^n) /. {x0 → .99, α → 1.3}) == .86, {n, 20}]
```

```
{n → 10.0588}
```

The result is 5 stages as we expected!

■ Variation of n with α.

```
alphaplot = Plot[
  n /. FindRoot[((1 - α^n + x0 α^n) /. {x0 → .99}) == .86, {n, 20}][[1]],
  {α, 1.01, 1.3}, PlotRange → {0, 300},
  AxesLabel → {"α", "number of trays"}]
```



We can guess the functional form:

```
guessplot =
 Plot[2.7 / (α - 1), {α, 1.01, 1.3}, PlotStyle → Red, PlotRange → All]
```




```
Show[alphaplot, guessplot]
```



Sort of amazing! Perhaps there is some explicit formula?

```
Solve[(1 - α^n + x0 α^n) == β, n]
```

— Solve:

Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

$$\left\{\left\{n \to \frac{\text{Log}\left[\frac{-1+\beta}{-1+x0}\right]}{\text{Log}[\alpha]}\right\}\right\}$$

Mathematica knows it! We see that the result is a version of the "Fenske" equation. (King, Separation Processes, 1971, p 456)

$$\text{guessplot2} = \text{Plot}\left[\left(\frac{\text{Log}\left[\frac{-1+\beta}{-1+x0}\right]}{\text{Log}[\alpha]}\right) /. \{\beta \to .86, x0 \to .99\}, \{\alpha, 1.01, 1.3\}, \text{PlotStyle} \to \text{Orange}, \text{PlotRange} \to \text{All}\right]$$



Sure enough it works!

```
Show[alphaplot, guessplot, guessplot2]
```



How could I guess so well?

$$\text{Series}\left[\frac{\text{Log}\left[\frac{-1+\beta}{-1+x0}\right]}{\text{Log}[\alpha]}, \{\alpha, 1, 2\}\right]$$

$$\frac{\text{Log}\left[\frac{-1+\beta}{-1+x0}\right]}{\alpha-1} + \frac{1}{2}\text{Log}\left[\frac{-1+\beta}{-1+x0}\right] - \frac{1}{12}\text{Log}\left[\frac{-1+\beta}{-1+x0}\right](\alpha-1) + \frac{1}{24}\text{Log}\left[\frac{-1+\beta}{-1+x0}\right](\alpha-1)^2 + O[\alpha-1]^3$$

What is the numerical value of the numerator!

$$\text{Log}\left[\frac{-1+\beta}{-1+x0}\right] \text{ /. } \{\beta \to .86, x0 \to .99\}$$

2.63906

Close to what I guessed from the intersection at α=1.01.

Have to say this is good fun!

Now what if want to allow for a finite reflux ratio?

For this case there is more work to do and the result is much less elegant.

The calculations for when the operating line is not the 45 degree seem to be best done in a loop. We pick "g" for the intersections with the equilibrium line (although we could use the complete relation if we wished.) and "f" for the intersections with the operating line.
Again we start at x0=y0=.99

```
g[0] = .99
```

```
0.99
```

```
f[0] = .99
```

```
0.99
```

Here is the loop for α=1.3. We do 10 steps. First for infinite reflux ratio.

Making operating line 45.

```
Do[g[i] = 1 - 1.3 + 1.3 g[i - 1];
 f[i] =  1 / 1 g[i] + (1 - 1) × .99, {i, 1, 10}]
```

The equilibrium intersections are:

```
ans11 = Table[{g[i], g[i - 1]}, {i, 1, 10}]
```

```
{{0.987, 0.99}, {0.9831, 0.987}, {0.97803, 0.9831},
 {0.971439, 0.97803}, {0.962871, 0.971439}, {0.951732, 0.962871},
 {0.937251, 0.951732}, {0.918427, 0.937251},
 {0.893955, 0.918427}, {0.862142, 0.893955}}
```

The operating line intersections are:

```
ans22 = Table[{g[i], f[i]}, {i, 1, 10}]
```

```
{{0.987, 0.987}, {0.9831, 0.9831}, {0.97803, 0.97803},
 {0.971439, 0.971439}, {0.962871, 0.962871},
 {0.951732, 0.951732}, {0.937251, 0.937251},
 {0.918427, 0.918427}, {0.893955, 0.893955}, {0.862142, 0.862142}}
```

As before

```
ans3 = Flatten[Table[{ans11⟦i⟧, ans22⟦i⟧}, {i, 1, Length[ans22]}], 1]
```

```
{{0.987, 0.99}, {0.987, 0.987}, {0.9831, 0.987},
 {0.9831, 0.9831}, {0.97803, 0.9831}, {0.97803, 0.97803},
 {0.971439, 0.97803}, {0.971439, 0.971439}, {0.962871, 0.971439},
 {0.962871, 0.962871}, {0.951732, 0.962871},
 {0.951732, 0.951732}, {0.937251, 0.951732}, {0.937251, 0.937251},
 {0.918427, 0.937251}, {0.918427, 0.918427}, {0.893955, 0.918427},
 {0.893955, 0.893955}, {0.862142, 0.893955}, {0.862142, 0.862142}}
```

Need the first point

```
Join[{{.99, .99}}, ans3]
```

```
{{0.99, 0.99}, {0.987, 0.99}, {0.987, 0.987},
 {0.9831, 0.987}, {0.9831, 0.9831}, {0.97803, 0.9831},
 {0.97803, 0.97803}, {0.971439, 0.97803}, {0.971439, 0.971439},
 {0.962871, 0.971439}, {0.962871, 0.962871}, {0.951732, 0.962871},
 {0.951732, 0.951732}, {0.937251, 0.951732}, {0.937251, 0.937251},
 {0.918427, 0.937251}, {0.918427, 0.918427}, {0.893955, 0.918427},
 {0.893955, 0.893955}, {0.862142, 0.893955}, {0.862142, 0.862142}}
```

```
test1 = ListPlot[%, Joined → True]
```




```
eqtest = Plot[(1 + (x - 1)/α) /. α → 1.3, {x, .86, 1}, PlotStyle → Green]
```




```
optest = Plot[((L x + G y0 - L y0)/G) /. {L → 1, G → 1, y0 → .99},
  {x, .86, .99}, PlotStyle → Red]
```




```
testans = Show[test1, eqtest, optest]
```



from above

```
%445
```



You can trust me that it matches above”

```
Show[testans, %445]
```



■ Making operating line an arbitrary value

L=.9, G=1

```
Clear[f, g]

g[0] = .99

0.99

f[0] = .99

0.99

Do[g[i] = 1 - 1.3 + 1.3 f[i - 1];
 f[i] =  .9 / 1 g[i] + (1 - .9) × .99, {i, 1, 10}]

ans111 = Table[{g[i], f[i - 1]}, {i, 1, 10}]

{{0.987, 0.99}, {0.98349, 0.9873}, {0.979383, 0.984141},
 {0.974578, 0.980445}, {0.968957, 0.976121},
 {0.962379, 0.971061}, {0.954684, 0.965142},
 {0.94568, 0.958216}, {0.935146, 0.950112}, {0.922821, 0.940631}}

ans222 = Table[{g[i], f[i]}, {i, 1, 10}]

{{0.987, 0.9873}, {0.98349, 0.984141}, {0.979383, 0.980445},
 {0.974578, 0.976121}, {0.968957, 0.971061},
 {0.962379, 0.965142}, {0.954684, 0.958216},
 {0.94568, 0.950112}, {0.935146, 0.940631}, {0.922821, 0.929539}}

ans31 =
 Flatten[Table[{ans111[[i]], ans222[[i]]}, {i, 1, Length[ans222]}], 1]

{{0.987, 0.99}, {0.987, 0.9873}, {0.98349, 0.9873},
 {0.98349, 0.984141}, {0.979383, 0.984141}, {0.979383, 0.980445},
 {0.974578, 0.980445}, {0.974578, 0.976121},
 {0.968957, 0.976121}, {0.968957, 0.971061}, {0.962379, 0.971061},
 {0.962379, 0.965142}, {0.954684, 0.965142}, {0.954684, 0.958216},
 {0.94568, 0.958216}, {0.94568, 0.950112}, {0.935146, 0.950112},
 {0.935146, 0.940631}, {0.922821, 0.940631}, {0.922821, 0.929539}}

ans41 = Join[{{.99, .99}}, ans31]
```

```
{{0.99, 0.99}, {0.987, 0.99}, {0.987, 0.9873},
 {0.98349, 0.9873}, {0.98349, 0.984141}, {0.979383, 0.984141},
 {0.979383, 0.980445}, {0.974578, 0.980445}, {0.974578, 0.976121},
 {0.968957, 0.976121}, {0.968957, 0.971061}, {0.962379, 0.971061},
 {0.962379, 0.965142}, {0.954684, 0.965142}, {0.954684, 0.958216},
 {0.94568, 0.958216}, {0.94568, 0.950112}, {0.935146, 0.950112},
 {0.935146, 0.940631}, {0.922821, 0.940631}, {0.922821, 0.929539}}
```

steps2 = ListPlot[ans41, Joined → True]



eq2plot = Plot$\left[\left(1 + \frac{x-1}{\alpha}\right)$ **/. α → 1.3, {x, .86, 1}, PlotStyle → Green**$\right]$



```
optest = Plot[((L x + G y0 - L y0)/G) /. {L → .9 , G → 1, y0 → .99},
  {x, .86, .99}, PlotStyle → Red]
```




```
Show[steps2, eq2plot, optest]
```



So we could only go from 0.92 to 0.99 at this L/G.

We can make it a little worse.

Making operating line an arbitrary value

L=.85, G=1

```
Clear[f, g]

g[0] = .99

0.99

f[0] = .99

0.99

Do[g[i] = 1 - 1.3 + 1.3 f[i - 1];
 f[i] =   .85 / 1 g[i] + (1 - .85) × .99, {i, 1, 10}]

ans111 = Table[{g[i], f[i - 1]}, {i, 1, 10}]

{{0.987, 0.99}, {0.983685, 0.98745}, {0.980022, 0.984632},
 {0.975974, 0.981519}, {0.971502, 0.978078},
 {0.966559, 0.974276}, {0.961098, 0.970075},
 {0.955063, 0.965433}, {0.948395, 0.960304}, {0.941026, 0.954636}}

ans222 = Table[{g[i], f[i]}, {i, 1, 10}]

{{0.987, 0.98745}, {0.983685, 0.984632},
 {0.980022, 0.981519}, {0.975974, 0.978078},
 {0.971502, 0.974276}, {0.966559, 0.970075}, {0.961098, 0.965433},
 {0.955063, 0.960304}, {0.948395, 0.954636}, {0.941026, 0.948372}}

ans31 =
 Flatten[Table[{ans111[[i]], ans222[[i]]}, {i, 1, Length[ans222]}], 1]

{{0.987, 0.99}, {0.987, 0.98745}, {0.983685, 0.98745},
 {0.983685, 0.984632}, {0.980022, 0.984632},
 {0.980022, 0.981519}, {0.975974, 0.981519}, {0.975974, 0.978078},
 {0.971502, 0.978078}, {0.971502, 0.974276}, {0.966559, 0.974276},
 {0.966559, 0.970075}, {0.961098, 0.970075}, {0.961098, 0.965433},
 {0.955063, 0.965433}, {0.955063, 0.960304}, {0.948395, 0.960304},
 {0.948395, 0.954636}, {0.941026, 0.954636}, {0.941026, 0.948372}}
```

```
ans41 = Join[{{.99, .99}}, ans31]
```

```
{{0.99, 0.99}, {0.987, 0.99}, {0.987, 0.98745},
 {0.983685, 0.98745}, {0.983685, 0.984632}, {0.980022, 0.984632},
 {0.980022, 0.981519}, {0.975974, 0.981519}, {0.975974, 0.978078},
 {0.971502, 0.978078}, {0.971502, 0.974276}, {0.966559, 0.974276},
 {0.966559, 0.970075}, {0.961098, 0.970075}, {0.961098, 0.965433},
 {0.955063, 0.965433}, {0.955063, 0.960304}, {0.948395, 0.960304},
 {0.948395, 0.954636}, {0.941026, 0.954636}, {0.941026, 0.948372}}
```

```
steps2 = ListPlot[ans41, Joined → True]
```



$$\texttt{eq2plot} = \texttt{Plot}\left[\left(1 + \frac{x - 1}{\alpha}\right) \texttt{/.}\ \alpha \to 1.3,\ \{x, .86, 1\},\ \texttt{PlotStyle} \to \texttt{Green}\right]$$




```
optest = Plot[((L x + G y0 - L y0)/G) /. {L → .85 , G → 1, y0 → .99},
  {x, .86, .99}, PlotStyle → Red]
```




```
Show[steps2, eq2plot, optest]
```



So we could only go from 0.94 to 0.99 at this L/G.

How does the result vary with α?

```
Clear[f, g]

Do[g[i] = 1 - α + α f[i - 1];
 f[i] =  m g[i] + (1 - m) y0, {i, 1, 10}]
```

We can again see the infinite reflux result for α-=1.3, ~.86 and as L/G decreases, the separation degree decreases rapidly.

```
Plot[f[10] /. {α → 1.3, f[0] → .99, y0 → .99},
 {m, .7, 1}, PlotLegends → {"α->1.3"},
 AxesLabel → {"L/G", "composition"}, PlotRange → {.85, 1}]
```




```
Plot[f[10] /. {α → 1.2, f[0] → .99, y0 → .99},
 {m, .7, 1}, PlotLegends → {"α->1.2"},
 AxesLabel → {"L/G", "composition"}, PlotStyle → Red]
```



```
Plot[f[10] /. {α → 1.1, f[0] → .99, y0 → .99},
 {m, .7, 1}, PlotLegends → {"α->1.1"},
 AxesLabel → {"L/G", "composition"}, PlotStyle → Green]
```



We see the substantial effect of reflux ratio and that the effect of α is most pronounced near infinite reflux ratio!

```
Show[%%%, %%, %]
```



Unfortunately, we can't make any progress with a solution for "n". We see that for the 5th tray, all powers of n appear,

```
Expand[f[5]]
```

$$m + y0 - m\,y0 - m\,\alpha + m^2\,\alpha + m\,y0\,\alpha - m^2\,y0\,\alpha - m^2\,\alpha^2 + m^3\,\alpha^2 + m^2\,y0\,\alpha^2 - m^3\,y0\,\alpha^2 - m^3\,\alpha^3 + m^4\,\alpha^3 + m^3\,y0\,\alpha^3 - m^4\,y0\,\alpha^3 - m^4\,\alpha^4 + m^5\,\alpha^4 + m^4\,y0\,\alpha^4 - m^5\,y0\,\alpha^4 - m^5\,\alpha^5 + m^5\,\alpha^5\,f[0]$$

With the previous result recovered:

```
% /. m → 1
```

$$1 - \alpha^5 + \alpha^5\,f[0]$$

- **So let's declare victory with the result that "n" varies as 1/(α-1) or more specifically as $\frac{\text{Log}\left[\frac{-1+\beta}{-1+x0}\right]}{\text{Log}[\alpha]}$!**